Calculations in k[x_1, x_2,...,x_n]/I where I is an ideal: finite dimensional case.

```
> interface(imaginaryunit=_i):
> with(Groebner):
> I := [x*y^3-x^2,x^3*y^2-y];
```

$$I := [x\,y^3 - x^2,\, x^3\,y^2 - y]$$

```
> G := Basis(I,tdeg(x,y));
```

$$G := [-x\,y + y^4,\, x\,y^3 - x^2,\, x^4 - y^2,\, x^3\,y^2 - y]$$

```
> LTI := map(LeadingMonomial,G,tdeg(x,y));
```

$$LTI := [y^4,\, x\,y^3,\, x^4,\, x^3\,y^2]$$

```
> LTC := [1,x,x^2,x^3,y,x*y,x^2*y,x^3*y,y^2,y^2*x,y^2*x^2,y^3];
```

$$LTC := [1,\, x,\, x^2,\, x^3,\, y,\, x\,y,\, y\,x^2,\, x^3\,y,\, y^2,\, x\,y^2,\, y^2\,x^2,\, y^3]$$

```
> nops({op(LTC)});
```

$$12$$

```
> f := 2*x+3*y^2-2;
```

$$f := 2\,x + 3\,y^2 - 2$$

```
> finv := add( c[i]*LTC[i], i=1..nops(LTC) );
```

$$finv := c_1 + c_2\,x + c_3\,x^2 + c_4\,x^3 + c_5\,y + c_6\,x\,y + c_7\,y\,x^2 + c_8\,x^3\,y + c_9\,y^2 + c_{10}\,x\,y^2 + c_{11}\,y^2\,x^2 + c_{12}\,y^3$$

```
> zero := NormalForm( f*finv, G, tdeg(x,y) ) - 1;
```

$$zero := -2\,c_1 + (-2\,c_2 + 2\,c_1)\,x + (-2\,c_5 + 2\,c_{11} + 3\,c_4)\,y + (-2\,c_{12} + 3\,c_5 + 2\,c_8)\,y^3 + (-2\,c_3 + 2\,c_2 + 2\,c_{12} + 3\,c_6)\,x^2 + (-2\,c_4 + 2\,c_3 + 3\,c_7)\,x^3 + (-2\,c_9 + 3\,c_1 + 2\,c_4 + 3\,c_8)\,y^2 + (3\,c_{12} + 2\,c_9 - 2\,c_{10} + 3\,c_2)\,x\,y^2 + (-2\,c_6 + 2\,c_5 + 3\,c_9)\,x\,y + (3\,c_{10} - 2\,c_7 + 2\,c_6)\,y\,x^2 + (3\,c_{11} - 2\,c_8 + 2\,c_7)\,x^3\,y + (-2\,c_{11} + 2\,c_{10} + 3\,c_3)\,y^2\,x^2 - 1$$

The main point here is that if f has an inverse $f^{(-1)}$ then $f f^{(-1)}$ = 1 mod I. Since I is finite dimensional and we know the complement of <LT(I)> then we know the form of elements in I so we can set up a linear system over k, which in this example is **Q**, to solve for the coefficients of $f^{(-1)}$.

```
> eqns := {coeffs(zero,[x,y])};
```

$$eqns := \{3\,c_{10} - 2\,c_7 + 2\,c_6,\, -2\,c_6 + 2\,c_5 + 3\,c_9,\, 3\,c_{12} + 2\,c_9 - 2\,c_{10} + 3\,c_2,\, -2\,c_3 + 2\,c_2 + 2\,c_{12} + 3\,c_6,\, -2\,c_9 + 3\,c_1 + 2\,c_4 + 3\,c_8,\, 3\,c_{11} - 2\,c_8 + 2\,c_7,\, -2\,c_{11} + 2\,c_{10} + 3\,c_3,\, -2\,c_4 + 2\,c_3 + 3\,c_7,\, -2\,c_{12} + 3\,c_5 + 2\,c_8,\, -2\,c_5 + 2\,c_{11} + 3\,c_4,\, -2\,c_2 + 2\,c_1,\, -2\,c_1 - 1\}$$

```
> sols := solve(eqns);
```

$$sols := \left\{c_6 = -\frac{4972}{51783},\, c_{12} = \frac{74503}{103566},\, c_{11} = \frac{9269}{51783},\, c_{10} = \frac{3416}{51783},\, c_9 = -\frac{13624}{51783},\, c_8 = \frac{28111}{103566},\right.$$

$$c_7 = \frac{152}{51783},\ c_5 = \frac{15464}{51783},\ c_4 = \frac{4130}{51783},\ c_3 = \frac{3902}{51783},\ c_2 = -\frac{1}{2},\ c_1 = -\frac{1}{2}\}$$

Now check it by multiplication in the quotient ring.

```
> NormalForm( f*subs(sols,finv), G, tdeg(x,y) );
```

$$1$$

```
> G2 := factor( Basis(I,plex(x,y)) );
```

$$G2 := [y\,(y-1)\,(y+1)\,(y^4+y^3+y^2+y+1)\,(y^4-y^3+y^2-y+1),\ y\,(-y^3+x),\ (-y^3+x)\,(x+y^3)]$$

Clearly y is a zero divisor in the quotient ring so it will not have an inverse.

```
> f := y;
```

$$f := y$$

```
> zero := NormalForm( f*finv, G, tdeg(x,y) ) - 1:
> eqns := {coeffs(zero,[x,y])};
```

$$eqns := \{-1,\ c_3,\ c_4,\ c_5,\ c_6,\ c_7,\ c_9,\ c_{10},\ c_{11},\ c_2+c_{12},\ c_1+c_8\}$$

```
> sols := solve(eqns);
```

$$sols :=$$

How many solutions are in V where V = **V**(I) ?

```
> Gy := Groebner[Basis](I,plex(x,y));
```

$$Gy := [y^{11}-y,\ x\,y-y^4,\ x^2-y^6]$$

```
> fy := Gy[1];
```

$$fy := y^{11}-y$$

```
> Gx := Groebner[Basis](I,plex(y,x));
```

$$Gx := [x^{12}-x^2,\ y-x^7]$$

```
> fx := Gx[1];
```

$$fx := x^{12}-x^2$$

It follows that the variety of I is finite, and there are at most 12 solutions for x and 11 for y hence |V| <= 132.

These univariate polynomials fx and fy which are in the ideal I can be found directly using the following command. One can compute them using the FGLM algorithm instead of using the lex monomial ordering to eliminate all variables except 1.

```
> Groebner[UnivariatePolynomial](x,I);
```

$$x^{12}-x^2$$

```
> Groebner[UnivariatePolynomial](y,I);
```

$$y^{11}-y$$

I'll count the number of solutions by computing a prime decomposition of the radical of I.

```
> with(PolynomialIdeals):
> I := <I>;
```

$$I := \langle x\,y^3-x^2,\ x^3\,y^2-y\rangle$$

```
> IsRadical(I);
```

$$false$$

The use of simplify below is to compute Groebner bases for each prime component of J.

```
> map(Simplify,[PrimeDecomposition(Radical(I))]);
```

$$[\langle y+1,\ x+1\rangle,\ \langle x^4+x^3+x^2+x+1,\ y-x^2\rangle,\ \langle x^4-x^3+x^2-x+1,\ y+x^2\rangle,\ \langle x,\ y\rangle,\ \langle y-1,\ x-1\rangle]$$

From here we see that there are 1 + 4 + 4 + 1 + 1 = 11 solutions.